
En la CAN-045 les presentamoslos módulostransceptoresTRW-2.4Gde Wenshing,transceivers que operan
en la banda de 2.4GHz, con capacidad de direccionamientoy selecciónde canalde comunicaciones.En esta
oportunidad,portamos elcódigo a Holtek, con lo cualpodemosemplearlospara implementareconómicos
remotos que se encargan de tomar una medición y reportarla al master cuando éste lo requiere, por ejemplo.

Desarrollo propuesto

Vamos aimplementaruna pequeñaespeciede biblioteca de funcionesque seocupe de enviar y recibir
mensajesutilizandoestosmódulos,paralo cual comenzaremos pordesarrollarrutinasparaescribiry leer un
byte en el módulo:

; Envía un byte
; dato en A
trw_write_byte:

mov datas,a ; guarda dato a transmitir
mov a,8 ; 8 bits
mov bitcntr,a
clr DATT ; DATA = output
nop

twbl1: rlc datas ; C = MSB (shift bit for output)
clr SCK ; CLK = 0
set DAT ; DATA = 1 (adivino)
snz C ; si hay carry, tuve suerte
clr DAT ; sino DATA = 0
nop
set SCK ; CLK = 1
sdz bitcntr
jmp twbl1
clr SCK ; CLK = 0
nop
set DATT ; DATA = input
ret

; Recibe un byte
; Retorna con dato en A
trw_read_byte:

clr datas ; DATA = 0
mov a,8 ; 8 bits
mov bitcntr,a

trbl1: set SCK ; CLK = 1
nop
rrca DATPORT ; lee dato en C
rlc datas ; y lo pone en el registro (MSB primero)
clr SCK ; CLK = 0
sdz bitcntr
jmp trbl1
mov a,datas
ret

Como habrándeducido,la selecciónde los pines a emplear serealizamediantemacros, las cuales pueden
redefinirse dentro del programa:

; DAT debe ser el LSB, de otro modo hay que modificar el programa
DAT  equ PG.0
DATPORT equ PG

DATT equ PGC.0
SCK equ PC.5
CE equ PB.3
DR equ PB.7
CS equ PB.2

; Tamaño de paquete en bytes
TRW_DATA_LEN equ 24
; Cantidad de bytes de dirección
TRW_ADDRESS_LEN equ 5

trwvars .section 'data'
datas db ?
bitcntr db ?
bytcntr db ?
addrptr db ?
; DEBE SER ESPACIO CONTIGUO
txbuffer LABEL BYTE
address db TRW_ADDRESS_LEN DUP(?)
messge db TRW_DATA_LEN DUP(?)

Luego seguimospor el procesode inicialización. La configuración deseada la guardamoscomo una tabla.
Comolas tablas nodebencruzarun espaciode 256programwords,en realidadla definimosen unaposición
fija y segura de memoria.

; Inicializa el TRW-2.4G
trw_init:

set CS ; CS = 1
clr bytcntr

til1: mov a,bytcntr
call getconfig ; obtiene byte de tabla en A
call trw_write_byte ; envía byte en A
inc bytcntr
mov a,18 ; 18 bytes
sub a,bytcntr
sz ACC
jmp til1 ; repite
clr CS ; CS = 0
ret

trwdata .section at LASTPAGE-0100h 'code'
; ...
getconfig:

addm a,PCL
ret a,08Eh
ret a,008h
ret a,01Ch
ret a,0C0h
ret a,0C0h
ret a,0BBh
ret a,0BBh
ret a,0BBh
ret a,0BBh
ret a,0BBh
ret a,013h ; LOCAL ADDRESS
ret a,025h ;
ret a,046h ;
ret a,079h ;
ret a,08Ah ;
ret a,0A3h
ret a,04Fh
ret a,014h

A continuación,un par de rutinasparaenviar y recibir un mensaje.Nótesecomo para insertarla dirección
antes del mensaje atransmitir, hemos definidolos dosbuffers contiguos. Elprocesode direccionamiento lo
hacemos copiando la direccióndesdeunatabla, la cual tambiéndebecumplir los requisitos mencionados en el
párrafoanterior.La rutina derecepciónindicasi vuelvecon las manos vacíaso trayendoun paquete, según el
valor de A, esto nos permitirá realizar otras tareas mientras esperamos un mensaje.

; Habilita tranmisión
trw_settx:

clr CE ; CE = 0

nop
set CS ; CS = 1
nop
clr DATT ; DATA = output
nop
clr DAT ; DATA = 0
nop
nop

sndbit: set SCK ; CLK = 1
nop
clr SCK ; CLK = 0
nop
set DATT ; DATA = input
nop
clr CS ; CS = 0
nop
set CE ; CE = 1
ret

; Habilita recepción
trw_setrx:

clr CE ; CE = 0
nop
set CS ; CS = 1
nop
clr DATT ; DATA = output
nop
set DAT ; DATA = 1
jmp sndbit

; Envía un mensaje
; Cargar previamente la dirección en address y el mensaje en message
trw_sendpacket:

call trw_settx
mov a,(TRW_ADDRESS_LEN+TRW_DATA_LEN) ; # bytes
mov bytcntr,a ; Inicializa # bytes a enviar
mov a,OFFSET txbuffer ; Inicializa puntero
mov MP0,a

tsl1: mov a,IAR0 ; obtiene byte en A
call trw_write_byte ; envía byte en A
inc MP0 ; siguiente
sdz bytcntr ; loop
jmp tsl1
clr CE ; CE = 0
ret

; Obtiene un mensaje
; Devuelve:
; A=0 => no recibió nada
; A=0xFF => hay un mensaje en el buffer
trw_getpacket:

snz DR ; packet ?
ret a,0 ; NO
clr CE ; CE = 0
mov a,TRW_DATA_LEN ; # bytes
mov bytcntr,a ; Inicializa # bytes a obtener
mov a,OFFSET messge ; Inicializa puntero
mov MP0,a

tgl1: call trw_read_byte ; obtiene byte en A
mov IAR0,a ; guarda byte en A en buffer
inc MP0 ; siguiente
sdz bytcntr ; loop
jmp tgl1
ret a,0FFh

; Setea dirección a quien transmitir
; Llamar con el número de remoto en A (0 en este caso, hay uno sólo cargado... el master...)
trw_address:

call addresses ; Obtiene posición donde está la dirección del
; remoto solicitado

mov addrptr,a ; la guarda
mov a,TRW_ADDRESS_LEN ; # bytes

mov bytcntr,a
mov a,OFFSET address ; Inicializa puntero
mov MP0,A

tal1: call getaddr ; obtiene address byte en A
mov IAR0,a ; lo guarda en el buffer
inc MP0 ; siguiente
inc addrptr
sdz bytcntr ; byte count
jmp tal1 ; loop
ret

trwdata .section at LASTPAGE-0100h 'code'

getaddr: mov a,addrptr
addresses:

addm a,PCL
ret a,LOW (addrs1-2) ; 2 bytes offset

; ret a,LOW (addrs2-2)
; ...
addrs1: ret a,021h

ret a,043h
ret a,065h
ret a,087h
ret a,0A9h

; addrs2
; ...
getconfig:
; ...

Dado que setrata de un medio de accesomúltiple en el cual no tenemosdetecciónde portadorani de
colisiones,es menesterarbitrar de algún modo la comunicaciónpara minimizar retransmisionesy lograr
confirmar que la información llega a destino. La forma más simplesueleser tenerun sistema configurado
comomaster,el cualinterrogaa los remotosy obtienerespuesta de ellos. Si lapreguntadel masterno llega a
destino,éstela retransmitirá.Si la respuesta del remoto no llega, elmasterretransmitirála pregunta.Los
remotosdeberán estar preparadospararecibir variasvecesla mismapregunta,sin ofenderse.Con estafilosofía
desarrollamosun pequeñoprogramitade ejemploqueanteunapetición contesta. Elmaster(quien pregunta)
en nuestrocasose realizó conun Rabbit,aprovechando la CAN-045, con la felizy pacíficacoexistenciade
ambosremotos,Rabbity Holtek. Nada impide, empero, que elavezadolectorconstruyasumasterconHoltek,
si así lo desea.

call TXdelay ; Esperar al arrancar antes de
call trw_init ; inicializar el módulo (devuelve A=0)
mov a,0 ; selecciona master
call trw_address ; Setea master address (para responder)

main: call trw_setrx ; Habilita rx
mw4msg: call trw_getpacket ; Obtiene msg

snz ACC.0 ; A!=0, sólo miro bit0
; (trw_getpacket devuelve 00 o FF)

jmp mw4msg ; no, loop
call TXdelay ; El master necesita un tiempo para habilitar Rx

; (Rabbit CAN-045 espera 3ms luego de Tx)
; acá va el proceso necesario para obtener la respuesta, si >3ms, no hace falta demora

mov a,041h ; sí, prepara ACK
mov messge,a
mov a,043h
mov messge[1],a
mov a,04Bh
mov messge[2],a
clr messge[3]
call trw_sendpacket ; y contesta
call TXdelay ; espera >3ms para que salga Tx
jmp main

Lasrutinasdesarrolladasfueron escritas teniendo en cuenta lasespecificacionesdel fabricante,para4MHz de
clock. La cantidadde callsanidadasse comprobóparaun HT48E30.Paraotras aplicaciones, se sugiere al
interesado revisar que secumplanlos tiemposdetallados en elmanualdel usuario delTRW-2.4Gy la cantidad
de calls permitidos por el hardware stack del micro de Holtek utilizado.
El archivo adjunto contiene la totalidad de los listados

